Problema 245 de *trianguloscabri*

Francisco Javier García Capitán

Resumen

Con el fin de resolver un problema de triángulos y dedicada a todos los asiduos de *trianguloscabri* hacemos una extensa introducción a las coordenadas baricéntricas. Después de resolver el problema planteado, usando *Mathematica* y los conceptos aprendidos, planteamos y resolvemos otras cuestiones relacionadas con dicho problema.

Contenido

1. El problema

1.1. Enunciado del problema

Juan Carlos Salazar propone el siguiente problema en *trianguloscabri*, la revista de triángulos de Ricardo Barroso Campos:

> *Sea P un punto interior del triángulo ABC, siendo $A_1B_1C_1$ su triángulo ceviano. Si trazamos un círculo tangente en A_1 a BC y tangente a la circunferencia circunscrita (O) de ABC, determinamos el punto de tangencia A_2 situado en el arco que no contiene a A. De manera similar definimos los puntos B_2 y C_2.*
>
> A) *Probar que AA_2, BB_2 y CC_2 son concurrentes.*
>
> B) *Si P es el punto de Gergonne, A_1A_2, B_1B_2 y C_1C_2 son concurrentes.*
>
> Taller de Olimpiadas de Vietnam

1.2. Trazado de la figura

Lo primero que queremos hacer es trazar la figura correspondiente al problema. Para ello tendremos en cuenta que si A_2 va a ser el punto de tangencia de la circunferencia circunscrita a ABC con la circunferencia referida en el enunciado, entonces A_2 será el *centro de semejanza directo* de las dos circunferencias, por lo que será el punto de intersección de la recta que une los centros y los extremos de dos radios paralelos trazados en el mismo sentido.



Por otro lado, al ser la circunferencia buscada tangente a BC en A_1, el centro O_2 de dicha circunferencia deberá estar en la perpendicular a BC por A_1.

Esto motiva la siguiente construcción: Trazamos el diámetro UV perpendicular a BC (U en el mismo arco que A, y V en el otro). La recta UA_1

cortará a la circunferencia circunscrita a ABC en A_2 y la recta OA_2 cortará a la perpendicular a BC por A_1 en el centro O_2 de la circunferencia buscada.

2. Coordenadas baricéntricas

Incluimos en esta sección los conceptos básicos sobre coordenadas baricéntricas que usaremos en la resolución del problema, aunque animamos al lector a que eche un buen vistazo a las referencias para estudiar las demostraciones y adquirir de esa manera un mayor control sobre el tema.

2.1. ¿Qué son las coordenadas baricéntricas?

Las coordenadas baricéntricas de un punto P son tres números u, v, w proporcionales a las áreas de los triángulos PBC, PCA, PAB. Por definición, los números u, v y w quedan determinados salvo un factor de proporcionalidad, por lo que los representaremos en la forma $u : v : w$.

Que el punto P tenga coordenadas $u : v : w$ equivale a decir que P se exprese algebraicamente según la relación $(u+v+w)P = uA+vB+wC$, que justifica el nombre de coordenadas baricéntricas. Observemos que la igualdad anterior exige que $u + v + w \neq 0$.



1. El *baricentro* G tiene coordenadas baricéntricas homogéneas $(1 : 1 : 1)$, ya que las áreas GBC, GCA y GAB son iguales.

2. El *incentro* I tiene coordenadas baricéntricas homogéneas $a : b : c$, ya que si r es el radio de la circunferencia inscrita, las áreas de los triángulos IBC, ICA e IAB son, respectivamente $\frac{1}{2}ar$, $\frac{1}{2}br$ y $\frac{1}{2}cr$.

3. El *circuncentro* O. Si R es el radio de la circunferencia circunscrita, las coordenadas de O son

$$\begin{aligned}
&\triangle OBC : \triangle OCA : \triangle OAB = \\
=&\tfrac{1}{2}R^2 \operatorname{sen} 2A : \tfrac{1}{2}R^2 \operatorname{sen} 2B : \tfrac{1}{2}R^2 \operatorname{sen} 2C = \\
=&\operatorname{sen} A \cos A : \operatorname{sen} B \cos B : \operatorname{sen} C \cos C = \\
=&a \cdot \frac{b^2+c^2-a^2}{2bc} = b \cdot \frac{c^2+a^2-b^2}{2ac} = c \cdot \frac{a^2+b^2-c^2}{2ab} = \\
=&a^2(b^2+c^2-a^2) : b^2(c^2+a^2-b^2) : c^2(a^2+b^2-c^2).
\end{aligned}$$

Notación: Si definimos

$$S_A = \frac{b^2+c^2-a^2}{2}, S_B = \frac{c^2+a^2-b^2}{2}, S_C = \frac{a^2+b^2-c^2}{2},$$

las coordenadas del circuncentro son $(a^2S_A : b^2S_B : c^2S_C)$.

Observemos aquí que $S_B + S_C = a^2$. También es posible comprobar que $a^2S_A + b^2S_B + c^2S_C = 4S^2$ llamando, como será habitual, S al doble del área del triángulo ABC. En efecto, teniendo en cuenta la fórmula de Herón para el área del triángulo y haciendo un poco de manipulación algebraica,

$$\begin{aligned}
&a^2(b^2+c^2-a^2) + b^2(c^2+a^2-b^2) + c^2(a^2+b^2-c^2) = \\
=&2a^2b^2 + 2a^2c^2 + 2b^2c^2 - a^4 - b^4 - c^4 = \\
=&(a+b+c)(-a+b+c)(a-b+c)(a+b-c) = \\
=&4 \cdot 4 \cdot \frac{a+b+c}{2} \cdot \frac{-a+b+c}{2} \cdot \frac{a-b+c}{2} \cdot \frac{a+b-c}{2} = \\
=&4S^2.
\end{aligned}$$

2.2. Área de un triángulo

Si las coordenadas baricéntricas homogéneas P, Q y R respecto del triángulo ABC son $P = (u_1 : v_1 : w_1)$, $Q = (u_2 : v_2 : w_2)$ y $R = (u_3 : v_3 : w_3)$, entonces,

$$\frac{(PQR)}{(ABC)} = \frac{\begin{vmatrix} u_1 & v_1 & w_1 \\ u_2 & v_2 & w_2 \\ u_3 & v_3 & w_3 \end{vmatrix}}{(u_1+v_1+w_1)(u_2+v_2+w_2)(u_3+v_3+w_3)}.$$

2.3. Ecuación de una recta

La fórmula del área de un triángulo nos sirve para razonar que la ecuación de la recta que pasa por los puntos $P = (u_1 : v_1 : w_1)$, $Q = (u_2 : v_2 : w_2)$ será

$$\begin{vmatrix} x & y & z \\ u_1 & v_1 & w_1 \\ u_2 & v_2 & w_2 \end{vmatrix} = 0,$$

que resultará una ecuación de la forma $px + qy + rz = 0$.

El lector comprobará sin dificultad que la ecuación de la recta BC es $x = 0$, y que la ecuación de la bisectriz interior del ángulo A es $cy - bz = 0$.

2.4. Punto de intersección de dos rectas

Las rectas de ecuaciones $p_1x+q_1y+r_1z = 0$ y $p_2x+q_2y+r_2z = 0$ se cortan en el punto $(q_1r_2 - q_2r_1 : r_1p_2 - r_2p_1 : p_1q_2 - p_2q_1)$, como puede comprobarse directamente en las ecuaciones, o resolviendo el sistema en función de un parámetro.

Es conveniente observar la simetría de las fórmulas que calculan la ecuación de una recta que pasa por dos puntos y el punto de intersección de dos rectas: *es exactamente la misma fórmula.*

2.5. Concurrencia y alineación

Teniendo en cuenta la fórmula del área del triángulo, es evidente que tres puntos estarán alineados si el determinante de sus coordenadas se anula; en realidad esto lo hemos usado para obtener la ecuación de una recta. Asimismo, tres rectas serán concurrentes si se anula el determinante de sus coeficientes. Obsérvese que el determinante

$$\begin{vmatrix} p_1 & q_1 & r_1 \\ p_2 & q_2 & r_2 \\ p_3 & q_3 & r_3 \end{vmatrix},$$

con los coeficientes de tres rectas por filas, puede entenderse, si lo desarrollamos por la primera fila, como el resultado de sustituir en la primera recta el punto de intersección de las otras dos.

2.6. División de un segmento en una razón dada

Si conocemos las coordenadas de dos puntos P y Q y ambas tienen la misma suma (es decir, ambos puntos tienen el mismo peso), el punto X de la recta PQ tal que $PX : XQ = m : n$ se calcula mediante la fórmula

$$(m+n)X = nP + mQ.$$

En particular, el punto medio de dos punto P y Q, cuyas coordenadas tengan la misma suma, se halla simplemente sumando las coordenadas de los dos puntos.

Observemos que, en cualquier caso, si los puntos dados no tienen la misma suma, siempre podremos multiplicar las coordenadas de cada punto por la suma de las del otro, y así conseguir que ambos tengan la misma suma.

Como ejemplo, hallemos las coordenadas baricéntricas del ortocentro H del triángulo ABC: el ortocentro está en la recta de Euler y divide al segmento OG externamente en la razón $3 : -2$. Hemos visto que

$$\begin{aligned} O &= \left(a^2(b^2+c^2-a^2) : b^2(c^2+a^2-b^2) : c^2(a^2+b^2-c^2)\right), \\ G &= (1,1,1), \end{aligned}$$

siendo 3 la suma de coordenadas de G y $4S^2$ la suma de las de O.

Lo primero que hacemos es multiplicar un factor adecuado las coordenadas de cada punto para que ambas sumen lo mismo, en este caso, $12S^2$:

$$\begin{aligned} O &= \left(3a^2(b^2+c^2-a^2) : 3b^2(c^2+a^2-b^2) : 3c^2(a^2+b^2-c^2)\right), \\ G &= (4S^2, 4S^2, 4S^2). \end{aligned}$$

La primera coordenada de H será

$$\begin{aligned} &(-2)3a^2(b^2+c^2-a^2) + 3\cdot 4S^2 = \\ =&-6(a^2b^2+a^2c^2-a^4) + 3(2a^2b^2+2a^2c^2+2b^2c^2-a^4-b^4-c^4) = \\ =&3a^4-3b^4+6b^2c^2-3c^4 = \\ =&3(a^2-b^2+c^2)(a^2+b^2-c^2) = 12S_BS_C \end{aligned}$$

Entonces obtnenemos que $H = (S_BS_C : S_CS_A : S_AS_B)$.

2.7. Puntos del infinito

En Geometría Moderna a cada recta se le añade un punto del ideal o punto del infinito. Es natural asignar a dicho punto coordenadas baricéntricas $(x : y : z)$ tales que $x + y + z = 0$, ya que si $x + y + z \neq 0$ obtenemos un punto real.

Todos los puntos del infinito están contenidos en una recta ideal o recta del infinito, que consideraremos con ecuación $x + y + z = 0$. Estos convenios cumplen los axiomas esperados: cada recta contiene un punto del infinito, dos rectas paralelas se cortan en un punto del infinito, etc.

El punto del infinito de la recta $px + qy + rz = 0$ se obtiene fácilmente hallando su intersección con $x+y+z = 0$, resultando el punto de coordenadas $(q - r : r - p : p - q)$.

Un resultado útil y sencillo de comprobar es que si $P = (u_1 : v_1 : w_1)$ y $Q = (u_2 : v_2 : w_2)$ son dos puntos cuyas coordenadas tienen la misma suma, entonces $Z = (u_1 - u_2 : v_1 - v_2 : w_1 - w_2)$ es el punto del infinito de la recta PQ. En efecto, sus coordenadas suman cero, y Z está alineado con P y Q.

2.8. Rectas perpendiculares

El cálculo de rectas perpendiculares presenta una de las primeras dificultades de las coordenadas baricéntricas. Como suele suceder, la solución de un problema difícil de una forma elegante hace tanto problema como a la solución el doble de atractivos.

Para hacerlo un poco más fácil resolveremos primero un caso simple y luego atacaremos el problema general.

Como lo que queremos es hallar la recta perpendicular a una recta dada por un punto dado, nos es suficiente encontrar el punto del infinito de dicha recta perpendicular.

2.8.1. Punto del infinito de la altura AH

La altura AH pasa por el vértice $A = (1 : 0 : 0)$ y el por el ortocentro $H = (S_B S_C : S_C S_A : S_A S_B)$, aunque también por la proyección A_H de H sobre BC. Como es $A_H = (0 : S_C S_A : S_A S_B) = (0 : S_C : S_B)$, con $S_B + S_C = a^2$, un punto del infinito de la recta AH será

$$(0 : S_C : S_B) - (a^2 : 0 : 0) = (-a^2 : S_C : S_B).$$

2.8.2. Punto del infinito de una perpendicular cualquiera

Dada una recta $\mathcal{L} : px + qy + rz = 0$, hallemos el punto del infinito de las rectas perpendiculares a ella. La recta $\mathcal{L}$ corta a las rectas CA y AB en los puntos $Y = (-r : 0 : p)$ y $Z = (q : -p : 0)$. Para hallar la perpendicular desde A a $\mathcal{L}$, primero hallaremos las ecuaciones de las perpendiculares desde Y a AB y desde Z a CA.



Estas son

$$\begin{vmatrix} S_B & S_A & -c^2 \\ -r & 0 & p \\ x & y & z \end{vmatrix} = 0, \quad \begin{vmatrix} S_C & -b^2 & S_A \\ q & -p & 0 \\ x & y & z \end{vmatrix} = 0,$$

o bien

$$\begin{aligned} &S_A px + (c^2 r - S_B p)y + S_A rz = 0, \\ &S_A px + S_A qy + (b^2 q - S_C p)z = 0. \end{aligned}$$

El punto de intersección de ambas rectas, ortocentro del triángulo AYZ, es

$$\begin{aligned} X' =&(*** : S_A p(S_A r - b^2 q + S_C p) : S_A p(S_A q + S_B p - c^2 r)) = \\ \sim&(*** : S_C(p-q) - S_A(q-r) : S_A(q-r) - S_B(r-p)), \end{aligned}$$

teniendo en cuenta que $S_A + S_C = b^2$ y $S_A + S_B = c^2$.

La perpendicular a $\mathcal{L}$ desde A es la recta AX', cuya ecuación es

$$\begin{vmatrix} 1 & 0 & 0 \\ *** & S_C(p-q) - S_A(q-r) & S_A(q-r) - S_B(r-p) \\ x & y & z \end{vmatrix} = 0,$$

o $-(S_A(q-r) - S_B(r-p))y + (S_C(p-q) - S_A(q-r))z = 0$.

Entonces, llamando $(f : g : h) = (q-r : r-p : p-q)$ al punto del infinito de la recta $\mathcal{L}$, la perpendicular a $\mathcal{L}$ desde A tiene ecuación

$$-(S_A f - S_B g)y + (S_C h - S_A f)z = 0,$$

con $(f' : g' : h') = (S_B g - S_C h : S_C h - S_A f : S_A f - S_B g)$ como punto del infinito, punto que pertenecerá a cualquier recta perpendicular a $\mathcal{L}$.

A modo de resumen, si $(f : g : h)$ es el punto del infinito de una recta ℓ, cualquier recta perpendicular a ℓ será paralela a $S_A fx + S_B gy + S_C hz = 0$, y el punto del infinito de cualquier recta perpendicular a ℓ será

$$(S_B g - S_C h : S_C h - S_A f : S_A f - S_B g).$$

3. Cálculos con *Mathematica*

A medida que pasemos de trabajar con puntos arbitrarios como $(u : v : w)$ en lugar de puntos concretos como $(0 : 1 : 1)$ la ayuda de un programa de cálculo simbólico como *Mathematica* se hará casi imprescindible.

Repasemos los conceptos vistos en la sección 2, pero ahora formulados como instrucciones de *Mathematica*.

La terna `{u,v,w}` representará tanto al punto de coordenadas $(u : v : w)$ como a la recta $ux + vy + wz = 0$.

Por ejemplo, introducimos directamente los vértices del triángulo y otros de los que sepamos las coordenadas de antemano: recordemos aquí que las líneas que unen el baricentro con los vértices forman tres triángulos de la misma área, y que en el caso del incentro, las áreas de los triángulos formados son $\frac{1}{2}ar$, $\frac{1}{2}br$, y $\frac{1}{2}cr$, siendo r el radio de la circunferencia inscrita. Entonces:

```
ptA = {1, 0, 0};
ptB = {0, 1, 0};
ptC = {0, 0, 1};
ptG = {1, 1, 1};
ptI = {a, b, c};
```

Dijimos que las coordenadas baricéntricas son únicas salvo un factor de proporcionalidad, de manera que, por ejemplo, el punto $a^4 : a^2b^2 : a^2c^2$ es equivalente a $a^2 : b^2 : c^2$. Podemos definir (y nos será muy útil) una función de *Mathematica* que efectúe este tipo de simplificaciones:

```
Simplificar[{x_, y_, z_}] := Simplify[{x, y, z} / PolynomialGCD[x, y, z]]
Simplificar[{a^4, a^2 b^2, a^2 c^2}]
```

```
{a^2, b^2, c^2}
```

También hemos visto que la fórmula que calcula las coordenadas del punto de intersección de dos rectas, o los coeficientes de la ecuación de una recta que pasa por dos puntos, es la misma, y precisamente es el producto vectorial, que *Mathematica* ya trae definido con la función `Cross`:

```
Punto[r_, s_] := Simplificar[Cross[r, s]]
Recta[P_, Q_] := Simplificar[Cross[P, Q]]
```

Ahora definimos la función `DividirRazon` que calcula el punto que divide a otros dos según una razón dada. Como dijimos en la sección 2, para usar la fórmula dada allí, es necesario que las coordenadas de ambos puntos tengan la misma suma, por lo que aquí multiplicamos cada punto por la suma de las coordenadas del otro. Para ello usamos la función *Traza* (`Tr`) de *Mathematica*. Después, usamos la función `DividirRazon` para para calcular el punto medio del segmento dado por dos puntos.

```
DividirRazon[P_, Q_, m_, n_] := Simplificar[n Tr[Q] P + m Tr[P] Q];
Medio[P_, Q_] := DividirRazon[P, Q, 1, 1];
```

A continuación, definimos las funciones `PuntoInfinito` para calcular el punto del infinito de la recta $px + qy + rz = 0$. A continuación, usamos dicha función para definir otra, `PuntoInfinitoPerpendicular` para obtener el punto del infinito de una recta cuyo punto del infinito sea $(f : g : h)$.

```
PuntoInfinito[{p_, q_, r_}] := {q - r, r - p, p - q};
SA = (b^2 + c^2 - a^2)/2; SB = (a^2 + c^2 - b^2)/2; SC = (a^2 + b^2 - c^2)/2;
PuntoInfinitoPerpendicular[{f_, g_, h_}] := PuntoInfinito[{SA f, SB g, SC h}]
```

Ya podemos hallar la recta perpendicular a una recta QR que pasa por un punto P:

```
Perpendicular[P_, Q_, R_] := Recta[P,
   PuntoInfinitoPerpendicular[PuntoInfinito[Recta[Q, R]]]];
```

Lo usamos para hallar la proyección ortogonal del punto P sobre la recta QR y la ecuación de la mediatriz del segmento PQ:

```
Pie[P_, Q_, R_] := Punto[Recta[Q, R], Perpendicular[P, Q, R]];
Mediatriz[P_, Q_] := Perpendicular[Medio[P, Q], P, Q];
```

Como aplicación, hallamos las coordenadas del circuncentro del triángulo ABC como intersección de dos mediatrices:

```
ptO = Punto[Mediatriz[ptB, ptC], Mediatriz[ptC, ptA]]
```

$\{a^2 (-a^2 + b^2 + c^2), b^2 (a^2 - b^2 + c^2), c^2 (a^2 + b^2 - c^2)\}$

4. Solución del problema

Vamos a resolver el problema sin tener que recurrir a la ecuacion en baricéntricas de la circunferencia circunscrita. Para ello usaremos que a) la prolongación de la bisectriz del ángulo A pasa por el punto V; b) el triángulo UA_2V es rectángulo en A_2, ya que UV es un diámetro.



Entonces, obtenemos el punto V como intersección de la bisectriz AI y la mediatriz del segmento BC. Como conocemos las coordenadas del circuncentro O, podemos hallar las del punto U, simétrico de V respecto de O. Para ello, tenemos en cuenta que $OU : UV = -1 : 2$.

```
ptP = {u, v, w};
ptA1 = Punto[Recta[ptA, ptP], Recta[ptB, ptC]]
ptV = Punto[Recta[ptA, ptI], Mediatriz[ptB, ptC]]
ptU = DividirRazon[ptO, ptV, -1, 2]
```

$\{0, v, w\}$

$\{a^2, -b\,(b+c), -c\,(b+c)\}$

$\{-a^2, b\,(b-c), c\,(-b+c)\}$

Como el lector atento habrá advertido, al no poner un punto y coma después de una instrucción, *Mathematica* muestra el resultado de dicha instrucción, mientras que si lo omitimos, *Mathematica* procesa la instrucción sin mostrar el resultado.

Ahora calculamos A_2 como proyección ortogonal de V sobre la recta A_2U:

```
ptA2 = Pie[ptV, ptU, ptA1]
```

$\{a^2\,v\,w, -b\,v\,(c\,v+b\,w), -c\,w\,(c\,v+b\,w)\}$

Para calcular B_2 y C_2 podemos repetir los mismos pasos, pero permutando los papeles de los vértices A, B y C. Mejor aún, podemos enseñar a *Mathematica* a realizar dicha permutación:

```
Permutar[{x_, y_, z_}] := {z, x, y} /.
  {a → b, b → c, c → a, u → v, v → w, w → u}
```

Ahora usamos `Permutar` para hallar B_1 y C_1, B_2 y C_2.

```
ptB1 = Permutar[ptA1]
ptC1 = Permutar[ptB1]
ptB2 = Permutar[ptA2]
ptC2 = Permutar[ptB2]
```

```
{u, 0, w}

{u, v, 0}

{-a u (c u + a w), b^2 u w, -c w (c u + a w)}

{-a u (b u + a v), -b v (b u + a v), c^2 u v}
```

El apartado A) de nuestro problema pide comprobar que las rectas AA_2, BB_2 y CC_2 son concurrentes:

```
Det[{
  Recta[ptA, ptA2],
  Recta[ptB, ptB2],
  Recta[ptC, ptC2]}]
```

```
0
```

El apartado B) afirma que si P es el punto de Gergonne del triángulo cuando ABC entonces A_1A_2, B_1B_2 y C_1C_2 son concurrentes.

El punto de Gergonne del triángulo ABC es el punto de intersección de las rectas AX, BY, CZ que se forman al unir los vértices del triángulo con los puntos de tangencia de la circunferencia inscrita con los lados opuestos.

Si usamos que $BX : XC = s - b : s - c$ y $CY : YA = s - c : s - a$, obtendremos fácilmente las coordenadas del punto de Gergonne:

```
s = (a + b + c)/2;
ptX = {0, s - c, s - b}; ptY = {s - c, 0, s - a};
ptGe = Punto[Recta[ptA, ptX], Recta[ptB, ptY]]
```

```
{a^2 - (b - c)^2, -a^2 + b^2 + 2 a c - c^2, -a^2 + 2 a b - b^2 + c^2}
```

El apartado B) parece sugerir que el punto de Gergonne no es el único que cumple la propiedad, así que preguntamos a *Mathematica* cuándo las rectas A_1A_2, B_1B_2 y C_1C_2 son concurrentes. Usamos la función `Factor` para que la expresión resultante aparezca factorizada, que será más fácil de interpretar:

```
Factor[Det[{
   Recta[ptA1, ptA2],
   Recta[ptB1, ptB2],
   Recta[ptC1, ptC2]}]]
```

```
a b c (-a b u + b^2 u + a c u - c^2 u - a^2 v + a b v - b c v + c^2 v + a^2 w - b^2 w - a c w + b c w)
  (c u v + b u w + a v w)
```

Vemos que hay dos factores que pueden anularse, uno de primer grado en u, v, w, correspondiente a una recta, y otro de segundo grado, que es una cónica.

Usamos la función `Part` de *Mathematica* para extraer la parte de la expresión correspondiente a la recta:

```
rectasol = Part[%, 4]
```

```
-a b u + b^2 u + a c u - c^2 u - a^2 v + a b v - b c v + c^2 v + a^2 w - b^2 w - a c w + b c w
```

Ahora bastará sustituir u, v y w por las coordenadas del punto de Gergonne y comprobar el apartado B) del problema. Para ello, y para usarla también más adelante, definimos una función que efectúe dicha sustitución.

```
Sustituir[f_, {x_, y_, z_}] := Simplify[f /. {u → x, v → y, w → z}]
```

Usamos la función que acabamos de definir con la expresión correspondiente a la recta y el punto de Gergonne:

```
Sustituir[rectasol, ptGe]
```

```
0
```

Luego, en efecto, el punto de Gergonne está en la recta y hemos acabado de comprobar el apartado B).

5. Otras cuestiones

Con los datos que hemos ido calculando y las funciones que hemos definido quizá podamos responder fácilmente a cuestiones relacionadas con el problema que acabamos de resolver.

A continuación planteamos y resolvemos algunas cuestiones, con la seguridad de que otras similares quedarán en el tintero, por lo que se invita al lector a continuar la tarea:

1. *¿Qué podemos decir del punto de intersección de las rectas AA_2, BB_2 y CC_2?*

```
Punto[Recta[ptA, ptA2],
      Recta[ptB, ptB2]]
```

```
{a u, b v, c w}
```

Se trata del llamado *producto baricéntrico* del incentro $I = (a : b : c)$ y el punto $P = (u : v : w)$.

2. *¿Hay algún punto notable más en la recta solución del apartado B)?*

```
Sustituir[rectasol, ptG]
```

```
0
```

En efecto, el baricentro G del triángulo ABC también está en la recta. Por tanto dicha recta queda caracterizada como la recta que pasa por el baricentro G y el punto de Gergonne Ge.

3. *Si P es un punto de la recta GG_e, ¿qué podemos decir del punto de intersección T de A_1A_2, B_1B_2 y C_1C_2?*

Podemos hallar T para $P = G$ y $P = G_e$

```
Sustituir[ptT, ptG]
Sustituir[ptT, ptGe]
```

$\{a^2\,(-a^2+b^2+c^2),\ b^2\,(a^2-b^2+c^2),\ c^2\,(a^2+b^2-c^2)\}$

$\{-a^2\,(-a+b+c)^2\,(a^4-(b-c)^4-2\,a^3\,c+2\,a\,(b-c)^2\,c),\ \ldots\}$

El primer punto nos es conocido, es el circuncentro. El segundo no tanto. Sin embargo, Juan Carlos Salazar, apunta que siempre T es un punto de la recta OI. Podemos comprobarlo con Cabri, y también podemos considerar un punto cualquiera de la recta GG_e y comprobamos que el punto T está alineado con O e I:

```
Det[{ptO, ptI, Sustituir[ptT, DividirRazon[ptG, ptGe, 1, λ]]}]
```

```
0
```

Es natural considerar los puntos de tangencia A_3, B_3, C_3 de las circunferencias trazadas internamente a los lados del triángulo ABC.

La construcción de estos nuevos puntos consiste simplemente en intercambiar los papeles de los puntos U y V.



En efecto, el ángulo VA_3U es recto y el centro O_3 de la nueva circunferencia estará en la perpendicular a BC por el punto A_1.

4. *Las rectas AA_3, BB_3 y CC_3 nunca son concurrentes.*

```
Factor[Det[{
    Recta[ptA, ptA3],
    Recta[ptB, ptB3],
    Recta[ptC, ptC3]}]]
```

$-2\,a\,b\,c\,u\,v\,w$

En efecto, esto solo ocurriría en el caso extremo de que el punto P estuviera en uno de los lados del triángulo, y no en su interior, como se está suponiendo.

5. *¿Cuándo son concurrentes las rectas A_1A_3, B_1B_3 y C_1C_3?*

```
Factor[Det[{
    Recta[ptA1, ptA3],
    Recta[ptB1, ptB3],
    Recta[ptC1, ptC3]}]]
```

$a\,b\,c\,(a+b+c)\,(-b\,c\,u^2\,v - c^2\,u^2\,v + a\,c\,u\,v^2 + c^2\,u\,v^2 + b^2\,u^2\,w + b\,c\,u^2\,w - a^2\,v^2\,w - a\,c\,v^2\,w - a\,b\,u\,w^2 - b^2\,u\,w^2 + a^2\,v\,w^2 + a\,b\,v\,w^2)$

Observamos que el lugar geométrico de P es una cúbica, que aparece descrita en la página de Bernard Gibert como $K_{317} = pK(X_{81}, X_{86})$, es decir una isocúbica pivotal con polo X_{81} y pivote X_{86} (los puntos están numerados de acuerdo con la enciclopedia de Clark Kimberling).

Como se observa en la figura de la página siguiente, esta curva pasa, entre otros puntos, por los vértices del triángulo, el baricentro G, el punto de Gergonne Ge, el incentro I y el punto de Schiffler S.



Alguna información sobre este tipo de curvas puede encontrarse la mencionada página de Bernard Gibert y en la de Eric Weisstein.

6. *Las rectas A_2A_3, B_2B_3 y C_2C_3 son siempre concurrentes.*

   En efecto,

```
Det[{
  Recta[ptA2, ptA3],
  Recta[ptB2, ptB3],
  Recta[ptC2, ptC3]}]
```

```
0
```

7. El punto de intersección de A_2A_3, B_2B_3 y C_2C_3 es el circuncentro O cuando P es el baricentro G y el punto simediano K cuando P es el incentro I.

```
ptN = Punto[
   Recta[ptA2, ptA3],
   Recta[ptB2, ptB3]];
Simplificar[Sustituir[ptN, ptG]]
Simplificar[Sustituir[ptN, ptI]]
```

$\{a^2(-a^2+b^2+c^2),\ b^2(a^2-b^2+c^2),\ c^2(a^2+b^2-c^2)\}$

$\{a^2,\ b^2,\ c^2\}$

Referencias

[1] http://www.personal.us.es/rbarroso/trianguloscabri
Página web Ricardo Barroso Campos, que dirige el Laboratorio Virtual de Triángulos.

[2] http://garciacapitan.auna.com/
Página web de Francisco Javier García Capitán. Puede descargarse de la sección *Escritos* una versión introductoria de la obra de Paul Yiu citada más abajo.

[3] http://perso.wanadoo.fr/bernard.gibert/
Página web de Bernard Gibert. Enciclopedia de curvas relacionadas con el triángulo.

[4] http://faculty.evansville.edu/ck6/encyclopedia/
Enciclopedia de puntos del triángulo de Clark kimberling.

[5] http://mathworld.wolfram.com/
Página web *Mathworld*, mantenida y creada por Eric Weisstein. Un sitio donde buscar términos relacionados con cualquier área de las Matemáticas.

[6] http://www.math.fau.edu/yiu/Geometry.html
Página web de Paul Yiu. Allí puede descargarse el monumental trabajo *Introduction to the Geometry of the Triangle, Summer 2001, version 2.0402* sobre geometría del triángulo con abundante uso de coordenadas baricéntricas.